
RTC Hodiny reálného času DS3231  

1. POPIS 

Modul DS3231 uchovává informace o aktuálním čase a datu. Podporuje 

časové údaje jako: sekundy, minuty, hodiny, dny, měsíce a roky, včetně 

přestupných let. Lze použít k velké škále modulů, které jsou závislé na 

časové informaci. Modul je schopný čas uchovat díky záložní baterii. 

Základní charakteristika: 

• 24hodinový nebo 12hodinový cyklus 

• Programovatelný budík a kalendář 

• Komunikační rozhraní I2C (IIC) 

• Pro baterie typu LIR2032 (není součástí balení) 

2. SPECIFIKACE  

Hlavní čip DS3231 Provozní napětí 3,3 až 5,5 V 

Paměťový čip AT24C32 Typ baterie  LIR2032 

Kapacita úložiště  32 KB Rozměry (mm) 38 x 22 x 14 

Odchylka kmitů oscilátoru < 2 ppm Hmotnost 8 g 

 

Poznámka:  

• ppm – parts-per million 

• S modulem lze použít jenom dobíjecí baterii LIR2032. V případě použití CR2032 může dojít 
k poškození baterie, výbuchu nebo požáru. 

3. ZAPOJENÍ 

 
 
 
 
 
 
 
 
 
 
 
 
 
 

4. UKÁZKA PROGRAMU 

Ukázka převzata z https://arduino.cz/tutorial-uzivani-hodin-realneho-casu-ds1307-a-ds3231-s-arduinem/ 

#include "Wire.h" 
#define DS3231_I2C_ADDRESS 0x68 
// Convert normal decimal numbers to binary coded decimal 
byte decToBcd(byte val){ 
    return( (val/10*16) + (val%10) ); 
} 
// Convert binary coded decimal to normal decimal numbers 
byte bcdToDec(byte val){ 
    return( (val/16*10) + (val%16) ); 
} 
void setup(){ 
    Wire.begin(); 
    Serial.begin(9600); 
    // set the initial time here: 
    // DS3231 seconds, minutes, hours, day, date, month, year 
    // setDS3231time(30,42,21,4,26,11,14); 
} 
void setDS3231time(byte second, byte minute, byte hour, byte dayOfWeek, byte dayOfMonth, byte month, byte year){ 
    // sets time and date data to DS3231 
    Wire.beginTransmission(DS3231_I2C_ADDRESS); 
    Wire.write(0); // set next input to start at the seconds register 
    Wire.write(decToBcd(second)); // set seconds 
    Wire.write(decToBcd(minute)); // set minutes 
    Wire.write(decToBcd(hour)); // set hours 
    Wire.write(decToBcd(dayOfWeek)); // set day of week (1=Sunday, 7=Saturday) 
    Wire.write(decToBcd(dayOfMonth)); // set date (1 to 31) 
    Wire.write(decToBcd(month)); // set month 
    Wire.write(decToBcd(year)); // set year (0 to 99) 
    Wire.endTransmission(); 
} 
void readDS3231time(byte *second, byte *minute, byte *hour, byte *dayOfWeek, byte *dayOfMonth, byte *month, byte *year){ 
    Wire.beginTransmission(DS3231_I2C_ADDRESS); 
    Wire.write(0); // set DS3231 register pointer to 00h 
    Wire.endTransmission(); 
    Wire.requestFrom(DS3231_I2C_ADDRESS, 7); 
    // request seven bytes of data from DS3231 starting from register 00h 
    *second = bcdToDec(Wire.read() & 0x7f); 
    *minute = bcdToDec(Wire.read()); 
    *hour = bcdToDec(Wire.read() & 0x3f); 
    *dayOfWeek = bcdToDec(Wire.read()); 
    *dayOfMonth = bcdToDec(Wire.read()); 
    *month = bcdToDec(Wire.read()); 
    *year = bcdToDec(Wire.read()); 
} 
void displayTime(){ 
    byte second, minute, hour, dayOfWeek, dayOfMonth, month, year; 

https://arduino.cz/tutorial-uzivani-hodin-realneho-casu-ds1307-a-ds3231-s-arduinem/

    // retrieve data from DS3231 
    readDS3231time(&second, &minute, &hour, &dayOfWeek, &dayOfMonth, &month, &year); 
    // send it to the serial monitor 
    Serial.print(hour, DEC); 
    // convert the byte variable to a decimal number when displayed 
    Serial.print(":"); 
    if (minute<10){ 
        Serial.print("0"); 
    } 
    Serial.print(minute, DEC); 
    Serial.print(":"); 
    if (second<10){ 
        Serial.print("0"); 
    } 
    Serial.print(second, DEC); 
    Serial.print(" "); 
    Serial.print(dayOfMonth, DEC); 
    Serial.print("/"); 
    Serial.print(month, DEC); 
    Serial.print("/"); 
    Serial.print(year, DEC); 
    Serial.print(" Day of week: "); 
    switch(dayOfWeek){ 
        case 1: 
            Serial.println("Sunday"); 
            break; 
        case 2: 
            Serial.println("Monday"); 
            break; 
        case 3: 
            Serial.println("Tuesday"); 
            break; 
        case 4: 
            Serial.println("Wednesday"); 
            break; 
        case 5: 
            Serial.println("Thursday"); 
            break; 
        case 6: 
            Serial.println("Friday"); 
            break; 
        case 7: 
            Serial.println("Saturday"); 
            break; 
    } 
} 

void loop(){ 
    displayTime(); // display the real-time clock data on the Serial Monitor, 
    delay(1000); // every second 
} 

 


